ÁLGEBRA Y GEOMETRÍA

Grado en ingeniería en Electrónica Industrial y Automática

Grado en ingeniería Eléctrica

Universidad de Cantabria

Práctica 7 (SAGE)

Aspectos generales de las prácticas de laboratorio:

- Abrir un navegador (recomendado Firefox o Chrome) y en la barra de dirección escribir

  193.146.75.191:8080

  Entrar con vuestras credenciales: Username (vuestro nombre con apellidos) y en Password, la contraseña.
- Crear la carpeta *practicas_algebra* en el escritorio y guardar todos los archivos realizados durante la clase en esa carpeta.
- Al finalizar la clase subir estos archivos a vuestro directorio en moodle y eliminar la carpeta *practicas_algebra* del escritorio.

Geometría Euclídea y aplicaciones

- **Producto escalar de vectores y norma de un vector**

```
-----------------------------------------------------------------------
u = vector(QQ, [1, 1, -1])
v = vector(ZZ, [1, 8, -2])
u.inner_product(v)
u.norm(); v.norm()
-----------------------------------------------------------------------
```

- **Procedimiento Gram Schmidt**

  Sea $\mathcal{S} = \{u_1, \ldots, u_r\}$ parte libre de un espacio euclídeo V y construimos $\mathcal{S}' = \{v_1, \ldots, v_r\}$:

$$v_i = u_i - \alpha_{i,i-1}v_{i-1} - \alpha_{i,i-2}v_{i-2} \ldots - \alpha_{i,2}v_2 - \alpha_{i,1}v_1, \qquad i = 1, \ldots, r, \qquad \alpha_{i,j} = \frac{u_i \cdot v_j}{v_j \cdot v_j}$$

  La primitiva *A.gram_schmidt*() convierte los vectores fila $\{u_1, \ldots, u_r\}$ de la matriz A en los vectores $\{v_1, \ldots, v_r\}$, también devuelve la matriz triangular inferior de los coeficientes $\alpha_{i,j}$.

```
-----------------------------------------------------------------------
A=matrix(QQ,[[1,1,0],[0,1,1],[1,0,1]])
GS=A.gram_schmidt()
GS[0], GS[1]
-----------------------------------------------------------------------
```

  El siguiente código de Sage tiene como entrada una matriz cuadrada inversible y, devuelve una matriz del mismo tamaño, donde las filas son la base ortonormal del procedimiento de Gram-Schmidt:

```
-----------------------------------------------------------------------
def Gram_Schmidt(A):
    G=matrix(RR,A.gram_schmidt()[0])
    for i in range(A.nrows()):
        G[i]=G[i]/G[i].norm()
    return G
-----------------------------------------------------------------------
```

Como es conocido, la matriz que devuelve el programa deberá ser ortogonal:

```
-----------------------------------------------------------------------
Gram_Schmidt(A)*Gram_Schmidt(A).transpose()
-----------------------------------------------------------------------
```

- **El método de los Mínimos cuadrados**

Sean $(x_1, y_1), (x_2, y_2), \cdots, (x_n, y_n)$ *un conjunto de n puntos de* $\mathbb{R}^2$ *tal que todos los* x_i *son distintos. Si*

$$A = \begin{pmatrix} 1 & x_1 & x_1^2 & \cdots & x_1^r \\ \vdots & \vdots & \vdots & & \\ 1 & x_n & x_n^2 & \cdots & x_n^r \end{pmatrix}$$

y

$$\begin{pmatrix} c_0 \\ \vdots \\ c_r \end{pmatrix} = (A^t A)^{-1} A^t \begin{pmatrix} y_1 \\ \vdots \\ y_n \end{pmatrix}$$

entonces $y = c_r x^r + \cdots + c_1 x + c_0$ *es el polinomio de grado r que da el mejor ajuste por mínimos cuadrados para los puntos considerados.*

```
-----------------------------------------------------------------------
def mc(KK,x,lis1,lis2,r):
    n=len(lis1)
    A=matrix(KK,n,r+1,[[z**i for i in range(r+1)] for z in lis1])
    C=(A.transpose()*A)**(-1)*A.transpose()*matrix(lis2).transpose()
    p=vector(C[i,0] for i in range(r+1))*vector([x**i for i in range(r+1)])
    return p
-----------------------------------------------------------------------
```

Ilustramos el programa con el conjunto de puntos considerados en la Práctica 5, sobre el polinomio de interpolación de la Lagrange:

```
-----------------------------------------------------------------------
puntos = [[-2,-7],[-1,12],[2,-3],[3,8]]; pp = list_plot(puntos); show(pp)
-----------------------------------------------------------------------
```

Computamos la recta que mejor ajuste a los puntos:

```
-----------------------------------------------------------------------
X=(-2,-1,2,3); Y=(-7,12,-3,8); p1=mc(QQ,x,X,Y,1); p1
-----------------------------------------------------------------------
```

Dibujamos

```
-----------------------------------------------------------------------
grafica1=plot(p1,-2.1,3.1)
show(grafica1+pp)
-----------------------------------------------------------------------
```

Ahora, computamos la parábola que mejor ajuste a los puntos:

```
-----------------------------------------------------------------------
p2=mc(QQ,x,X,Y,2);  grafica2=plot(p2,-2.1,3.1)
show(grafica2+pp)
-----------------------------------------------------------------------
```

¿ Cuál es el polinomio de grado 3 que mejor ajuste a los puntos ?

```
-----------------------------------------------------------------------
p4=mc(QQ,x,X,Y,3);  grafica4=plot(p4,-2.1,3.1)
show(grafica4+pp)
-----------------------------------------------------------------------
```

Lo comparamos con el Polinomio de Interpolación de Lagrange, implementado en la Práctica 5:

```
-------------------------------------------------------------------------
p=PolyVan(QQ,x,X,Y); p
-------------------------------------------------------------------------
```

- **Resolución de sistemas de ecuaciones lineales sobredimensionados**

Genéricamente los sistemas de ecuaciones lineales con más ecuaciones que incógnitas no tienen ninguna solución salvo que se verifiquen las hipótesis del Teorema de Rouche-Frobenius. Las técnicas desarrolladas en este capítulo nos van a permitir calcular, como en la sección anterior, la mejor pseudosolución, esto es el punto de $\mathbb{R}^n$ que está más proximo de ser una solución del sistema lineal considerado.

Teorema 1. *Sean A una matriz con m filas y n columnas, $m \geq n$ y $\underline{b}$ un vector columna en $\mathbb{R}^m$. Si*

$$\underline{x} = (A^t A)^{-1} A^t \underline{b}$$

entonces

$$\|A\underline{x} - \underline{b}\| < \|A\underline{y} - \underline{b}\|$$

para todo $\underline{y} \in \mathbb{R}^n$ tal que $\underline{y} \neq \underline{x}$.

La clave de la demostración de este teorema se encuentra en observar que el vector $\underline{x}$ verificando la condición exigida es justamente la proyección de $\underline{b}$ sobre el subespacio de $\mathbb{R}^m$ definido por las columnas de A.

Ilustramos este resultado con el siguiente sistema de ecuaciones lineales:

$$\left\{ \begin{array}{rcr} x+y+z & = & 1 \\ x-y-z & = & 0 \\ -x-y+2z & = & -1 \\ x-3y+5z & = & 2 \end{array} \right\}$$

El sistema $A\underline{x} = \underline{b}$, la matriz $A = \begin{pmatrix} 1 & 1 & 1 \\ 1 & -1 & -1 \\ -1 & -1 & 2 \\ 1 & -3 & 5 \end{pmatrix}$ tiene rango 3, y el rango de la matriz ampliada $A|b$ es 4.

Luego el sistema no tiene solución. La pseudosolución es $(A^tA)^{-1}A^t\underline{b} = \left(\; x = \frac{233}{268}, \quad y = \frac{47}{268}, \quad z = \frac{39}{134} \; \right)$, donde

$$(A^tA)^{-1}A^t = \begin{pmatrix} \frac{79}{268} & \frac{17}{67} & -\frac{22}{67} & \frac{33}{268} \\ \frac{93}{268} & -\frac{19}{67} & -\frac{3}{67} & -\frac{29}{268} \\ \frac{23}{134} & -\frac{13}{67} & \frac{5}{67} & \frac{13}{134} \end{pmatrix}$$

- **Factorización QR**

Dada una matriz real $A \in \mathbb{M}_{m\times n}(\mathbb{R})$, con columnas linealmente independientes, podemos encontrar matrices $Q \in \mathbb{M}_{m\times n}(\mathbb{R})$ y $R \in \mathbb{M}_{n\times n}(\mathbb{R})$ tales que

1. *$A = QR$.*
2. *Los vectores columnas de Q son ortonormales.*
3. *R es triangular superior inversible.*

La primitiva de Sage *A.QR()* devuelve las matrices Q y R verificando las propiedades de arriba, con coeficientes en *RDF*

```
-------------------------------------------------------------------------
A=matrix(RDF,[[1,1,0],[0,1,1],[1,0,1]])
FactQR=A.QR()
FactQR[0], FactQR[1]
(
[ -0.7071067811865472 -0.40824829046386296  -0.5773502691896257]
[                -0.0   -0.816496580927726   0.5773502691896257]
[ -0.7071067811865475    0.408248290463863   0.5773502691896257],

[ -1.4142135623730951  -0.7071067811865472  -0.7071067811865475]
[                 0.0   -1.224744871391589 -0.40824829046386313]
[                 0.0                  0.0   1.1547005383792515]
)
-------------------------------------------------------------------------
```

- **Ejercicios**

1. Aplica el procedimiento de Gram-Schmidt a los vectores $(1,-2,3,5),(7,-4,-2,5),(1,2,0,6)$ de $\mathbb{R}^4$.
2. En cada uno de los siguientes casos encuentra la recta que se ajusta mejor, y el ajuste cuadrático para los puntos dados.
   a) $(2,-5)$, $(3,0)$, $(1,1)$, $(4,-2)$
   b) $(-7,3)$, $(2,8)$, $(1,5)$
3. Demuestra que el sistema siguiente no tiene solución y, encuentra una pseudosolución:

$$\begin{cases} x+y & = & 1 \\ x-y & = & 0 \\ -x+3y & = & -1 \\ x-2y & = & 2 \end{cases}$$

4. Encontrar la factorización QR de la matriz

$$\begin{pmatrix} -2 & 0 & -1 & 2 & 4 \\ -1 & 0 & 1 & -1 & 2 \\ 0 & -2 & -2 & 0 & -1 \\ -1 & -1 & -1 & 0 & -1 \\ -1 & -1 & -2 & 1 & 0 \end{pmatrix}.$$